

Refined Ludus EBNF - Narrative-Focused

```
story          = "Story", string, { character }, { scene }, "End Story" ;

character      = "Character", identifier, ":", { char_attribute } ;

char_attribute = identifier, ":", value ;

scene         = "Scene", string, ":", { scene_content } ;

scene_content = narration | action | feeling | dialogue | choice | transition ;

narration     = "narrate:", string ;

action        = identifier, action_type ;

action_type   = "enters" | "appears" | "disappears" ;

feeling       = identifier, "feels:", value ;

dialogue      = identifier, "says:", string ;

transition    = "transition to", string, "with:", narration ;

choice        = "Choice:", { choice_option }, [ default_option ] ;

choice_option = "option", string, [ condition ], ":",
                { choice_body } ;

choice_body   = narration | variable_change | transition ;

default_option = "default:", choice_body ;

condition     = "if", identifier, ".", identifier, comparator, value ;

variable_change = identifier, change_type, identifier, ":", string ;

change_type   = "shows" | "loses" ;

comparator    = "==" | "!=" | "<" | ">" | "<=" | ">=" ;

value         = string | number | boolean ;

string        = "'" , { character_literal } , "'" ;

number        = digit , { digit } ;

boolean       = "true" | "false" ;

identifier    = letter , { letter | digit | "_" } ;

character_literal = ? printable character except quote ? ;
```

```
digit      = "0".."9" ;
```

```
letter     = "A".."Z" | "a".."z" ;
```